

--- Apr.16, 2012 ---

Manual for subroutines DC3D0 and DC3D to calculate displacement, strain and tilt at depth due to a point/rectangular strike/dip/tensile source in a half-space

Reference

Okada, Y., 1992, Internal deformation due to shear and tensile faults in a half-space,
Bull. Seism. Soc. Am., 82, 1018-1040.

DC3D0 : Internal deformation due to a point source

(1) Calling sequence


CALL DC3D0 (ALPHA, X, Y, Z, DEPTH, DIP, POT1, POT2, POT3, POT4,
UX, UY, UZ, UXX, UYX, UZX, UXY, UYY, UZY, UXZ, UYZ, UZZ, IRET)

Input arguments (REAL*4)

ALPHA : Medium constant, $(\lambda + \mu) / (\lambda + 2\mu) = 1 - (V_s/V_p)^2$, [V_p , V_s = P-, S-velocity]
 X, Y, Z : Coordinate of observation point (Z should be $Z \leq 0$)
 DEPTH : Depth of point source (DEPTH ≥ 0)
 DIP : Dip-angle of the source (degree)
 POT1 : Potency for strike-slip fault = (moment of double-couple) / $\mu = LWU$
 POT2 : Potency for dip-slip fault = (moment of double-couple) / $\mu = LWU$
 POT3 : Potency for tensile fault = (moment of isotropic part of dipole) / λ
 POT4 : Potency for explosive source = (moment of dipole) / μ

Output arguments (REAL*4)

UX, UY, UZ : Displacement
 UXX, UYX, UZX : X-derivative
 UXY, UYY, UZY : Y-derivative
 UXZ, UYZ, UZZ : Z-derivative
 IRET : Return code
 : =0...normal
 : =1...singular point
 : =2...positive Z was given



(IRET=1 occurs when the observation point coincides to the source position)

(2) Unit of output

Unit for UX, UY, UZ = (Unit of potency) / (Unit of X, Y, Z, DEPTH)**2
 Unit for UXX.. UZZ = (Unit of potency) / (Unit of X, Y, Z, DEPTH)**3

So, if potency is given in unit of cm^3 and X, Y, Z, DEPTH are given in unit of km,
 UX, UY, UZ (in cm) can be obtained by multiplication of $10^{(-10)}$ to the output, while
 UXX, UYX, ..., UZZ in correct unit can be obtained by multiplication of $10^{(-15)}$.

(3) Subroutines called by DC3D0 : DCCON0, DCCON1, UA0, UB0, UC0

DC3D : Internal deformation due to a rectangular fault source

(1) Calling sequence

CALL DC3D (ALPHA, X,Y,Z, DEPTH, DIP, AL1,AL2, AW1,AW2, DISL1,DISL2,DISL3,
UX, UY, UZ, UXX, UYX, UZX, UXY, UYY, UZY, UXZ, UYZ, UZZ, IRET)

Input arguments (REAL*4)

ALPHA : Medium constant, $(\lambda + \mu) / (\lambda + 2\mu) = 1 - (V_s/V_p)^2$, [$V_p, V_s = P-, S\text{-velocity}$]

X, Y, Z : Coordinate of observation point (Z should be $Z \leq 0$)

DEPTH : Depth of reference point on the fault surface (DEPTH ≥ 0)

DIP : Dip-angle of the fault surface (degree)

AL1, AL2 : Coordinate range in strike direction of the fault surface


AW1,AW2 : Coordinate range in up-dip direction of the fault surface
taking reference point as the origin of the fault surface

DISL1 : Dislocation in strike-slip component

DISL2 : Dislocation in dip-slip component

DISL3 : Dislocation in tensile component


If $0 < \text{DIP} < 90$, positive DISL1 gives left-lateral movement
and positive DISL2 gives reverse-fault movement
while positive DISL3 gives open-type movement



strike-slip



dip-slip



tensile

Output arguments (REAL*4)

UX, UY, UZ : Displacement

UXX,UYX, UZX : X-derivative

UXY, UYY, UZY : Y-derivative


UXZ, UYZ, UZZ : Z-derivative

IRET : Return code

: =0...normal

: =1...singular point

: =2...positive Z was given



(IRET=1 occurs when the observation point lies on the fault edges)

(2) Unit of output


Unit for UX,UY,UZ = (Unit of DISL)

Unit for UXX..UZZ = (Unit of DISL) / (Unit of X,Y,..,AW2)


So, if dislocation is given in unit of cm and X, Y, Z, DEPTH, AL1, AL2, AW1, AW2
are given in unit of km, UX,UY,UZ(in cm) can be obtained by output itself, while
UXX, UYX,..,UZZ in correct unit can be obtained by multiplication of $10^{**(-5)}$.

(3) Subroutines called by DC3D : DCCON0, DCCON2, UA, UB, UC

A sample of parameters
AL1=0, AL2=30, AW1=-20,AW2=0,
DEPTH=10, DIP=25, DISL2=50



By adopting AL1, AL2, AW1, AW2,
this subroutine can be easily
applied to inhomogeneous source,
i.e. a source composed by multi
segment faults, each of which may
have different dislocations.




Appendix-1 : Fault model in a geographical coordinate system

A geographical coordinate system may be adopted in actual application.

In such a case, follow the steps below.

- (1) Select reference point on the fault surface as (ALAT0, ALNG0).
- (2) Convert station's coordinate, (ALAT, ALNG) to EW and NS position, (XEW, YNS) using the subroutine, PRTOXY.
- (3) Convert (XEW, YNS) to the position (X, Y) in the fault coordinate system according to the strike direction.



```

SUBROUTINE PRTOXY ( ALATDG, ALNGDG, ALATO, ALNGO, X, Y, IND )
C
C*****
C***** Conversion between (lat, long) and (X, Y) *****
C***** using Gauss-Krueger projection *****
C*****
C*****
C*****
C
C***** Input/Output
C***** ALATDG, ALNGDG : latitude, longitude (deg)
C***** X, Y : EW, NS coordinates (km)
C
C***** Input
C***** ALATO, ALNGO : origin of projection (deg)
C***** IND = 0 ... transform (lat, lng) to ( X, Y )
C***** : 1 ... transform ( X, Y ) to (lat, lng)
C
parameter ( A=6378.160, E2=6.6946053E-3, E12=6.7397251E-3 )
parameter ( D=57.29578, RD=1./57.29578 )
C-----
C----- IND=0 : transform (lat, lng) to (X, Y) -----
C-----
IF (IND .EQ. 0) THEN
  RLAT = ALATDG*RD
  SLAT = SIN( RLAT )
  CLAT = COS( RLAT )
  V2 = 1. + E12*CLAT*CLAT
  AL = ALNGDG - ALNGO
  PH1 = ALATDG + 0.5*V2*AL*AL*SLAT*CLAT*RD
  RPH1 = PH1*RD
  RPH2 = (PH1 + ALATO)/2.*RD
  SRPH1 = SIN( RPH1 )
  SRPH2 = SIN( RPH2 )
C-----
  R = A*(1. - E2) / SQRT( 1. - E2*SRPH2*SRPH2 )**3
  AN = A / SQRT( 1. - E2*SRPH1*SRPH1 )
  C1 = D / R
  C2 = D / AN
  Y = (PH1 - ALATO) / C1
  X = AL*CLAT/C2*( 1. + AL*AL*COS( 2.*RLAT ) / ( 6.*D*D ) )
C-----
C----- IND=1 : transform (X, Y) to (LAT, LNG) -----
C-----
ELSEIF (IND .EQ. 1) THEN
  RLATO = ALATO*RD
  SLATO = SIN( RLATO )

```

```

      CLATO = COS ( RLATO )
      DEN = SQRT ( 1. - E2*SLATO*SLATO )
      R = A*( 1. - E2 ) / DEN**3
      AN = A / DEN
      V2 = 1. + E12*CLATO*CLATO
C-----
      C1 = D / R
      C2 = D / AN
      PH1 = ALATO + C1*Y
      RPH1 = PH1*RD
      TPH1 = TAN (RPH1)
      CPH1 = COS (RPH1)
      BL = C2*X
      ALATDG = PH1 - 0.5*BL*BL*V2*TPH1*RD
      ALNGDG = ALNGO+BL/CPH1*(1. - BL*BL*(1. +2.*TPH1*TPH1)/(6.*D*D))
      ENDIF
C-----
      RETURN
      END

```

Appendix-2 : Test routine to check displacement calculation by DC3D

```

C----- Check DC3D
      X=10.
      Y=20.
      Z=-30.
      DEPTH=50.
      DIP=70.
      AL1=-80.
      AL2=120.
      AW1=-30.
      AW2=25.
      DISL1=200.
      DISL2=-150.
      DISL3=100.
C-----
      WRITE (6, *) ' *** OUTPUT OF DC3D ***'
      WRITE (6, *)
      WRITE (6, *) ' DEPTH, DIP=' , DEPTH, DIP
      WRITE (6, *) ' AL1, AL2, AW1, AW2=' , AL1, AL2, AW1, AW2
      WRITE (6, *) ' DISL1, DISL2, DISL3=' , DISL1, DISL2, DISL3
      WRITE (6, *) ' X=' , X, ' Y=' , Y, ' Z=' , Z
C-----
      ALPHA=2. /3.
      CALL DC3D (ALPHA, X, Y, Z, DEPTH, DIP,
*              AL1, AL2, AW1, AW2, DISL1, DISL2, DISL3,
*              UX, UY, UZ, UXX, UYX, UZX, UXY, UYY, UZY, UXZ, UYZ, UZZ, IRET)
      WRITE (6, *)
      WRITE (6, *) ' IRET=' , IRET
      WRITE (6, *) ' UX, UY, UZ=' , UX, UY, UZ
      WRITE (6, *) ' ANSWER = -37.8981 63.1789 14.9607'
      STOP
      END

```

Yoshimitsu OKADA

Natl.Res.Inst. for Earth Sci. & Disas.Prev.
 Tennodai 3-1, Tsukuba, Ibaraki, 305-0006 JAPAN
 TEL : +81-29-863-7771, FAX : +81-29-863-7770
 e-mail : okada@bosai.go.jp